The (GAMS) Transport Problem

Outline

- Introduction
- Basic LP programmes
  - The diet problem
  - Comparative advantage
- The GAMS Transport Problem
  - Standard algebraic presentation
- Structure of a GAMS Programme
- The Transport Problem in GAMS Code
- Next

Introduction

- A classic linear programming (LP) problem
  - LP and CGE problems are optimisation problems
  - LP problems are a slightly simpler starting point
  - AN LP problem can demonstrate all the key elements in a GAMS programme
- The GAMS tutorial uses this LP programme
  - A printed copy of the GAMS tutorial may prove helpful.

Basic LP Programmes: Diet

- The diet problem
  - OBJ: minimise the cost (C) of achieving a minimum consumption of three nutrients (Z_1,Z_2,Z_3)
  - STO: the two available food commodities (X_1,X_2) supplying the nutrients in different ratio ($a_{i,j}$)

$$\text{Min } C = p_1.X_1 + p_2.X_2$$

$$\text{sto}$$

$$a_{11}.X_1 + a_{12}.X_2 \geq Z_1$$
$$a_{21}.X_1 + a_{22}.X_2 \geq Z_2$$
$$a_{31}.X_1 + a_{32}.X_2 \geq Z_3$$

Basic LP Programmes: Diet



Practical CGE, 2017 © cgemod 5

Basic LP Programmes: Comparative Adv.



Practical CGE, 2017 © cgemod 6

The GAMS Transport Problem

Minimise total transport costs of supplying three markets from two production plants



The GAMS Transport Problem

Indices/Sets

i = plants

j = markets

Available Data

a_i = supply of commodity at plant i (in cases)

b_j = demand for commodity at market j (in cases)

d_{ij} = distances between plant i and market j ($/mile//case)

f = freight cost ($/case/1,000 miles)

Decision Variables

X_{ij} = amount of commodity to ship from plant i to market j (cases)

The GAMS Transport Problem

Constraints

Supply limit at plant *i*: $\sum_j X_{ij} \leq a_i \qquad \forall i$

Demand at market *j*: $\sum_i X_{ij} \geq b_j \qquad \forall j$

$X_{ij} \geq 0 \qquad \forall i, j$

Objective Function

Minimise $\sum_i \sum_j c_{ij} X_{ij}$

The GAMS Transport Problem

Data

		Markets		
Plants	**New York**	**Chicago**	**Topeka**	**Supplies**
		(Distances '000 m)		
Seattle	2.5	1.7	1.8	350
San Diego	2.5	1.8	1.4	600
Demands	325	300	275	

Freight Cost

$90 per case per 1,000 miles

Structure of a GAMS Programme

SETS	Declaration
	Assignment of Members
Data (PARAMETERS, TABLES, SCALARS)	Declaration
	Assignment of Values
VARIABLES	Declaration
	Assignment of Type
	(optional) Assignment of bounds/initial values
EQUATIONS	Declaration
	Definition
MODEL and SOLVE statements	
(optional) DISPLAY statements	

Practical CGE, 2017 © cgemod 11

The Transport Problem in GAMS Code

```
$TITLE  A TRANSPORTATION PROBLEM (TRNSPORT,SEQ=1)
$OFFUPPER
* This problem finds a least cost shipping schedule that meets
* requirements at markets and supplies at factories

   SETS
       i   canning plants    / SEATTLE, SAN-DIEGO /
       j   markets           / NEW-YORK, CHICAGO, TOPEKA / ;

  PARAMETERS

       a(i)  capacity of plant i in cases
         /    SEATTLE      350
              SAN-DIEGO    600  /

       b(j)  demand at market j in cases
         /    NEW-YORK     325
              CHICAGO      300
              TOPEKA       275  / ;
```

Practical CGE, 2017 © cgemod 12

The Transport Problem in GAMS Code

```
TABLE d(i,j)  distance in thousands of miles
                  NEW-YORK       CHICAGO      TOPEKA
    SEATTLE          2.5            1.7          1.8
    SAN-DIEGO        2.5            1.8          1.4   ;

SCALAR f  freight in dollars per case per thousand miles  /90/ ;

PARAMETER c(i,j)  transport cost in '000 of dollars per case ;

          c(i,j) = f * d(i,j) / 1000 ;

VARIABLES
     X(i,j)  shipment quantities in cases
     Z       total transportation costs in thousands of dollars ;

POSITIVE VARIABLE X ;
```

Practical CGE, 2017 © cgemod 13

The Transport Problem in GAMS Code

```
EQUATIONS
     COST         define objective function
     SUPPLY(i)    observe supply limit at plant i
     DEMAND(j)    satisfy demand at market j ;

COST..         Z  =E=  SUM((i,j), c(i,j)*X(i,j)) ;

SUPPLY(i)..    SUM(j, X(i,j))  =L=  a(i) ;

DEMAND(j)..    SUM(i, X(i,j))  =G=  b(j) ;

MODEL TRANSPORT /ALL/ ;

SOLVE TRANSPORT USING LP MINIMIZING Z ;

DISPLAY X.L, X.M ;
```

Practical CGE, 2017 © cgemod 14

Next

- Transport Problem Exercises
- Exploring the transport problem model
- Debugging a GAMS model
  - Syntax errors
  - Execution errors
- Changing the model
  - Changing unit transport costs
  - Changing distances
  - Adding a new markets
  - Adding intermediate (wholesale) markets

The End